/*
 * Flashlib - QB9 common Flash library
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

package com.qb9.flashlib.easing
{
	import com.qb9.flashlib.tasks.ITask;
	
	/** Tween a set of attributes (usually x, y) to form a curved trajectory. */
	public class QuadBezierTween extends Tween
	{
		protected var control:Object;
		
		public function QuadBezierTween(obj:Object, duration:int, dest:Object, control:Object, options:Object=null)
		{
			super(obj, duration, dest, options);
			
			this.control = control;
		}
		
		protected override function tweenedAttribute(attr:String, tween:Number):Number
		{
			var v1:Number = src[attr];
			var v2:Number = control[attr];
			var v3:Number = dest[attr];
			var r:Number = 1 - tween;
			
			return (r * r * v1) + (2 * tween * r * v2) + (tween * tween *v3);
		}
		
		override public function clone():ITask
		{
			return new QuadBezierTween(obj, duration, dest, control, options);
		}
		
	}
}
